Problem:
Score : 300 points

Problem Statement

We will call a string obtained by arranging the characters contained in a string a in some order, an anagram of a.

For example, greenbin is an anagram of beginner. As seen here, when the same character occurs multiple times, that character must be used that number of times.

Given are N strings s_1, s_2, \ldots, s_N. Each of these strings has a length of 10 and consists of lowercase English characters. Additionally, all of these strings are distinct. Find the number of pairs of integers i, j (1 \leq i < j \leq N) such that s_i is an anagram of s_j.

Constraints

2 \leq N \leq 10^5

s_i is a string of length 10.

Each character in s_i is a lowercase English letter.

s_1, s_2, \ldots, s_N are all distinct.

Input

Input is given from Standard Input in the following format:

N
s_1
s_2
:
s_N

Output

Print the number of pairs of integers i, j (1 \leq i < j \leq N) such that s_i is an anagram of s_j.

Sample Input 1

3
acornistnt
peanutbomb
constraint

Sample Output 1

1

s_1 = acornistnt is an anagram of s_3 = constraint. There are no other pairs i, j such that s_i is an anagram of s_j, so the answer is 1.

Sample Input 2

2
oneplustwo
ninemodsix

Sample Output 2

0

If there is no pair i, j such that s_i is an anagram of s_j, print 0.

Sample Input 3

5
abaaaaaaaa
oneplustwo
aaaaaaaaba
twoplusone
aaaabaaaaa

Sample Output 3

4

Note that the answer may not fit into a 32-bit integer type, though we cannot put such a case here.

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program name
    use,intrinsic :: iso_fortran_env
    implicit none
    integer(int32):: i,j,n
    integer(int64):: ans
    character(10),allocatable:: s(:)
    character(1):: tmp
    logical:: ok

    read*, n
    allocate(s(n))
    do i=1,n
        read*, s(i)
    end do

    do i=1,n
        ok=.false.
        do while(.not. ok)
            ok = .true.
            do j=1,9
                if (s(i)(j:j) > s(i)(j+1:j+1)) then
                    ok = .false.
                    tmp = s(i)(j:j)
                    s(i)(j:j) = s(i)(j+1:j+1)
                    s(i)(j+1:j+1) = tmp
                end if
            end do
        end do
    end do

    call char_sort(s,1,n)
    j=1
    ans=0
    do i=1,n-1
        if (s(i) == s(i+1)) then
            ans=ans+j
            j=j+1
        else
            j=1
        end if
    end do

    print'(i0)', ans
contains
    recursive subroutine char_sort(c,fst,lst)
        integer(int32),intent(in):: fst, lst
        character(*),intent(inout):: c(:)
        integer(int32):: mdl

        if (lst-fst < 2)then
            if (c(fst) > c(lst)) call cswap(c(fst),c(lst))
            return
        end if

        mdl = (fst+lst)/2
        call char_sort(c,fst,mdl)
        call char_sort(c,mdl+1,lst)
        call char_merge(c,fst,mdl,lst)
    end subroutine

    subroutine char_merge(c,fst,mdl,lst)
        integer(int32),intent(in):: fst,mdl,lst
        character(*),intent(inout):: c(:)
        character(10),allocatable:: tmp(:)
        integer(int32):: l,r,ti

        allocate(tmp(lst-fst+1))
        l=fst
        r=mdl+1
        ti=1

        do while(l <= mdl .and. r <= lst)
            if (c(l) <= c(r)) then
                tmp(ti) = c(l)
                l=l+1
            else
                tmp(ti) = c(r)
                r=r+1
            end if
            ti=ti+1
        end do

        if (l <= mdl) then
            tmp(ti:) = c(l:mdl)
        else
            tmp(ti:) = c(r:lst)
        end if

        c(fst:lst) = tmp(:)
        deallocate(tmp)
    end subroutine

    subroutine cswap(c1,c2)
        character(*):: c1,c2
        character(10),allocatable:: tmp

        tmp = c1
        c1 = c2
        c2 = tmp
    end subroutine
end program name